
CRT 
Display 

  

(HP-41CX, Hewlett Packard 1983 and DM41X, SwissMicros 2020) 

 

Overview1 

The CRT program represents a solution to the Chinese Remainder Theorem which states that a linear 
system of congruence equations with pairwise relatively prime moduli has a unique solution modulo the 
product of the moduli of the system.  

Practically, it goes like this. Imagine a basket of eggs for which it is not known how many there are. Too 
lazy to count all of them one may know that if you take out three at a time, it ends up with two left-over. 
If one takes out five at a time, three are left-over, and by taking out seven at a time, two are left-over. This 
is enough information to figure out the least number of eggs that are in the basket. Here we go. 

The basis for finding a solution for an integer number x which satisfies the congruences of modulo 
definitions as explained below. 
 
Suppose that m1,m2,..,mk are pairwise relatively prime positive integers, and a1,a2,..,ak is a series of 
integers, congruences exist as follows: 

      x ≡ ai (mod mi)       for i=1..k and have a unique solution: 

 

      M = m1·m2·..·mk        which is given by: 

 

      x ≡ a1M1y1 + a2M2y2 + .. + akMkyk (mod M) with: 

 

      Mi = M/mi    and    yi ≡ 1/Mi (mod mi)    for i=1..k 

 

The values   yi  can be found by applying the Extended Euclidean Algorithm. 

 

 

Example 1 
Please note that my default FIX 5 setting which can be replaced by your preferred number of decimals 

at line 178. An example is used for the following three congruences: 

 

 
1 This program is copyright and is supplied without representation or warranty of any kind. The author assumes no responsibility and shall have no 
liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof 

https://www.swissmicros.com/

      x ≡ 2 (mod 3) 

      x ≡ 3 (mod 5) 

      x ≡ 2 (mod 7) 

 

in which M=LCM(3,5,7)=105 and the value of x is to be determined. 
 

KEYSTROKES DISPLAY COMMENTS 

  

Run CRT from the start with: 

     x ≡ 2 (mod 3) 

      x ≡ 3 (mod 5) 

      x ≡ 2 (mod 7) 

[XEQ] [ALPHA]CRT[ALPHA]  Enter m1, which is 3 

3[R/S]  Enter m2, which is 5 

5[R/S]  Enter m3, which is 7 

7[R/S]  Press [R/S] to stop more mi entries 

[R/S]  Enter a1, which is 2 

2[R/S]  Enter a2, which is 3 

3[R/S]  Enter a3, which is 2 

2[R/S]  Shows x 

[USER] [E]  Press User Key E to show M 

[USER] [D]  Press User Key D to show m1-m3 

[USER] [C]  Enter a value for X, e.g. 145 

145[R/S]  Values for a1-a3 for same set m1-m3 

[USER] [B]  Try these reverse values with a1=1 

1[R/S]  Enter a2=0 

0[R/S]  Enter a3=5 

5[R/S]  Shows lowest value for x 

[USER] [B]  Try these reverse values with a1=-1 

-1[R/S]  Enter a2=-1 

-1[R/S]  Enter a3=-1 

-1[R/S]  Shows lowest value for x 

[USER] [E]  Shows M 

[USER] [A]  Run again by entering new values for m1-mk 

 

Example 2 

Another example is given with four congruences: 
 
      x ≡  5 (mod  7) 
      x ≡  7 (mod 11) 
      x ≡ 14 (mod 31) 
      x ≡  8 (mod 45) 

in which M=LCM(7,11,31,45)=107415 and the value of x is to be determined. 

KEYSTROKES DISPLAY COMMENTS 

  

Run CRT from the start with: 

     x ≡  5 (mod  7) ≡  7 (mod 11) 

      x ≡ 14 (mod 31) ≡  8 (mod 45) 
[XEQ] [ALPHA]CRT[ALPHA]  Enter m1, which is 7 

7[R/S]  Enter m2, which is 11 

11[R/S]  Enter m3, which is 31 

31[R/S]  Enter m4, which is 45 

45[R/S]  Press [R/S] to stop more mi entries 

[R/S]  Enter a1, which is 5 

5[R/S]  Enter a2, which is 7 

7[R/S]  Enter a3, which is 14 

14[R/S]  Enter a4, which is 8 

8[R/S]  Shows x 

[USER] [E]  Press User Key E to show M=LCM(7,11,31,45) 

[USER] [D]  Press User Key D to show m1-m4 

[USER] [C]  Enter a value for X, e.g. 45555 

145[R/S]  Values for a1-a4 for same set m1-m4 

[USER] [B]  Try these reverse values with a1=6 

6[R/S]  Enter a2=4 

4[R/S]  Enter a3=16 

16[R/S]  Enter a4=15 

15[R/S]  Shows lowest value for x (same as entry) 

[USER] [A]  Try a new one with m1=3: 
      x ≡  1 (mod  3) ≡  6 (mod  7) 
      x ≡ 10 (mod 11) ≡  2 (mod 13) 
      x ≡ 13 (mod 17) ≡ 16 (mod 19) 
 3[R/S]  Enter m2=7 

7[R/S]  Enter m3=11 

11[R/S]  Enter m4=13 

13[R/S]  Enter m5=17 

17[R/S]  Enter m6=19 

19[R/S]  Continue to enter values ai 

[R/S]  Enter a1=1 

1[R/S]  Enter a2=6 

6[R/S]  Enter a3=10 

10[R/S]  Enter a4=2 

2[R/S]  Enter a5=13 

13[R/S]  Enter a6=16 

16[R/S]  Shows x 

[USER] [E]  Press User Key E for M=LCM(3,7,11,13,17,19) 

[USER] [A]  Run again by entering new values for m1-mk 

Program Listing 

The listing of CRT is given below with functions A-E in User Mode. 

 01▪LBL "CRT" 

 02▪LBL A 

 03 3 

 04 XEQ 10 

 05 1 
 06 ENTER 

 07▪LBL 02 

 08 "M" 

 09 ARCL X 

 10 >"=?" 

 11 PROMPT 

 12 FC?C 22 

 13 GTO 03 

 14 STO IND T 

 15 ST* Z 

 16 RDN 

 17 STO 01 
 18 1 

 19 ST+ T 

 20 + 

 21 GTO 02 

 22▪LBL 03 

 23 X<>Y 
 24 STO 02 

 25▪LBL B 

 26 XEQ 10 

 27 RCL 01 

 28 3 

 29 + 
 30 RCL 01 

 31 1 

 32▪LBL 05 

 33 "A" 

 34 ARCL X 

 35 >"=?" 

 36 PROMPT 

 37 FC?C 22 

 38 GTO C 

 39 STO IND T 

 40 RDN 

 41 1 
 42 ST+ T 

 43 + 

 44 X<=Y? 

 45 GTO 05 

 46 2 

 47 R^ 
 48▪LBL 08 

 49 RDN 

 50 1 

 51 ST+ Y 

 52▪LBL 06 

 53 STO Z 

 54 RCL 02 

 55 * 
 56 RCL IND Y 

 57 / 

 58 LASTX 

 59 MOD 

 60 1 

 61 X=Y? 

 62 GTO 07 

 63 X<>Y 

 64 RDN 

 65 RCL Z 

 66 + 

 67 GTO 06 
 68▪LBL 07 

 69 RDN 

 70 RDN 

 71 RCL X 

 72 RCL 01 

 73 ST+ X 
 74 + 

 75 R^ 

 76 STO IND Y 

 77 2 

 78 R^ 

 79 STO Z 
 80 - 

 81 RCL 01 

 82 + 

 83 X>0? 

 84 GTO 08 

 85 . 

 86 STO 00 

 87 2 

 88 R^ 

 89▪LBL 09 

 90 RDN 

 91 1 
 92 + 

 93 RCL 01 

 94 RCL 02 

 95 RCL IND Z 

 96 / 

 97 RDN 
 98 + 

 99 RCL IND X 

100 R^ 

101 * 

102 X<>Y 

103 RCL 01 

104 + 

105 RCL IND X 
106 X<>Y 

107 RDN 

108 * 

109 ST+ 00 

110 RDN 

111 STO Y 

112 2 

113 X<>Y 

114 - 

115 RCL 01 

116 + 

117 X>0? 
118 GTO 09 

119 RCL 00 

120 RCL 02 

121 MOD 

122 "X=" 

123 ARCL X 
124 XEQ 11 

125 PROMPT 

126 GTO B 

127▪LBL C 

128 "X=?" 

129 PROMPT 
130 STO 00 

131 XEQ 12 

132 "A=(" 

133 XEQ 10 

134▪LBL 00 

135 RCL 00 

136 RCL IND Y 

137 MOD 

138 ARCL X 

139 >"," 

140 RDN 

141 ISG X 
142 GTO 00 

143 XEQ 13 

144 PROMPT 

145 GTO C 

146▪LBL D 

147 "M=(" 
148 XEQ 12 

149 XEQ 10 

150▪LBL 14 

151 ARCL IND X 

152 RCL 02 

153 RCL IND Y 

154 ST/ Y 

155 MOD 
156 X=0? 

157 >"*" 

158 X<>Y 

159 >"," 

160 ISG X 

161 GTO 14 

162 XEQ 13 

163 PROMPT 

164 GTO D 

165▪LBL E 

166 XEQ 10 

167 "M*=" 
168 ARCL 02 

169 XEQ 11 

170 PROMPT 

171 GTO E 

172▪LBL 10 

173 CF 29 
174 FIX 00 

175 RTN 

176▪LBL 11 

177 SF 29 

178 FIX 05 

179 RTN 
180▪LBL 12 

181 RCL 01 

182 2 

183 + 

184 1 E3 

185 / 

186 3 

187 + 

188 RTN 

189▪LBL 13 

190 -1 

191 AROT 
192 ATOX 

193 >")" 

194 XEQ 11 

195 END 

 

 
 

 

(321 bytes) 

Registers, Labels and Flags 

REGISTERS COMMENTS  LABELS COMMENTS 

R00 Work solution for x  LBL00 Calculate and show ai 

R01 Number of mi and ai, k  LBL01 - 

R02 LCM(m1,..,mk)  LBL02 Entry of mi 

R03..R03+k Values m1..mk  LBL03 Point after entry mi 

R03+k+1..R03+2k Values a1..ak  LBL04 - 

   LBL05  Entry of ai 

   LBL06 Calculate yi and M  

   LBL07 Loop around yi and M 

   LBL08 Start of yi and M loop 

   LBL09 Calculate x 

   LBL10 Reset flag and fix number 

   LBL11 Set flag and fix number 

   LBL12 Set loop value for mi and ai 

   LBL13 Show mi and ai values 

   LBL14 Show value M 

   LBL A User Mode: Entry of all mi 

   LBL B User Mode: Entry of all ai 

   LBL C User Mode: Entry x 

   LBL D User Mode: Show all mi 

   LBL E User Mode: Show M 

  

FLAGS COMMENTS 

22 Check for keyboard input 

29 Thousands separator 

 
 

References 

Mathematical article: Chinese Remainder Theorem by University of Colorado Denver. 
Interactive math website: Chinese Remainder Theorem by Cut the Knot.  

 

Downloads 

The RAW/TXT format of the program is available via the website: CRT (in zip file). 

 

http://www-math.ucdenver.edu/~wcherowi/courses/m5410/ctccrt.html
https://www.cut-the-knot.org/blue/chinese.shtml
https://hooxies.nl/?page_id=28727

